# $Id: PKGBUILD 87533 2013-04-02 17:43:50Z bpiotrowski $
# Maintainer:  Bartłomiej Piotrowski <barthalion@gmail.com>
# Contributor: Thomas Dziedzic < gostrc at gmail >
# Contributor: Bert Muennich <muennich at informatik.hu-berlin.de>
# Contributor: Brad Fanella <bradfanella@archlinux.us>

pkgname=sxiv
pkgver=1.1
pkgrel=2
pkgdesc='Simple X Image Viewer'
arch=('i686' 'x86_64')
license=('GPL2')
install=sxiv.install
url='https://github.com/muennich/sxiv'
depends=('libx11' 'xproto' 'imlib2' 'giflib' 'desktop-file-utils')
source=(https://github.com/muennich/sxiv/archive/$pkgver.tar.gz
        sxiv.desktop
        config.h)
sha256sums=('7d22b060d264be25ee0e95afff9df535299dbdd48067c9fc795133e583e2583f'
            'bd14b9b6bc6d78f18cdff8535b4d1c13b3c76e35091dd6bc675011d42a01bb47'
            '61515a87692849c942bfc3cce8b4c0f19139b4a1e9b6375a994cc541c64aab5c')

build() {
  cd sxiv-$pkgver
  cp "$srcdir"/config.h .
  make
}

package() {
  cd sxiv-$pkgver

  make PREFIX="$pkgdir"/usr install
  install -D -m644 "$srcdir"/sxiv.desktop \
                   "$pkgdir"/usr/share/applications/sxiv.desktop
}
